

PROGRAMMING LANGUAGES:

IMPERATIVE PROGRAM CONSTRUCTION

6. LOOP CONSTRUCTION II: STRENGTHENING THE INVARIANT

Shin-Cheng Mu

Autumn, 2024

National Taiwan University and Academia Sinica

MAXIMUM SEGMENT SUM

A classical problem: given an array of integers, find largest possible sum of a consecutive segment.

`con $N : \text{Int}$ $\{0 \leq N\}$`

`con $f : \text{array } [0..N)$ of Int`

`S`

`$\{r = \langle \uparrow p\ q : 0 \leq p \leq q \leq N : \text{sum } p\ q \rangle\}$`

where `$\text{sum } p\ q = \langle \sum i : p \leq i < q : f[i] \rangle$` .

DETAILS THAT MATTER

- Note the use of \leq and $<$ in the specification.
- The range in $\text{sum } p \ q$ is $p \leq i < q$. It computes the sum of $f[p..q)$ — not including $f[q]$!
- Therefore when $p = q$, $\text{sum } p \ q$ computes the sum of an empty segment.
- In the postcondition we have $p \leq q$ — we allow empty segments in our solution!
- We must have $q \leq N$ instead of $q < N$. Otherwise segments containing the rightmost element would not be valid solutions.

PREVIOUSLY INTRODUCED TECHNIQUES

- Replace N by n . Use $P \wedge Q$ as the invariant, where

$$\begin{aligned} P &\equiv r = \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle , \\ Q &\equiv 0 \leq n \leq N . \end{aligned}$$

- Use $\neg (n = N)$ as guard. This way we immediately have that $P \wedge Q \wedge n = N$ imply the desired postcondition.
- How do we know we want $0 \leq n \leq N$? It can be forced by our development later. But let's expedite the pace.
- Initialisation: $n, r := 0, 0$.
- Use $N - n$ as the bound.
- To decrease the bound, let $n := n + 1$ be the last statement of the loop.

We get this program.

```
con  $N : \text{Int}$   $\{0 \leq N\}$   
con  $f : \text{array } [0..N)$  of  $\text{Int}$   
var  $r, n : \text{Int}$   
  
 $r, n := 0, 0$   
 $\{P \wedge Q, \text{bnd} : N - n\}$   
do  $n \neq N \rightarrow ??? ; n := n + 1$  od  
 $\{r = \langle \uparrow p \ q : 0 \leq p \leq q \leq N : \text{sum } p \ q \rangle\}$ 
```

Now we need to construct the `???` part.

CONSTRUCTING THE LOOP BODY

How to construct the ??? part?

$$\{P \wedge Q \wedge n \neq N\}$$

???

$$\{(P \wedge Q)[n \setminus n + 1]\}$$
$$n := n + 1$$
$$\{P \wedge Q\}$$

CONSTRUCTING ASSIGNMENTS

How do you construct such an assignment?

$$\{r = \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle \wedge \\ Q \wedge n \neq N\}$$

$$r := ???$$

$$\{(P \wedge Q)[n \setminus n + 1]\}$$

$$n := n + 1$$

$$\{P \wedge Q\}$$

Recall what we have learnt: if from $(P \wedge Q)[n \setminus n + 1]$ we can infer that

$$r = \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle \oplus E ,$$

the statement $???$ could be $r := r \oplus E$.

SPLITTING OFF?

Let us look at the step “split off $q = n + 1$ ” in more detail:

$$\begin{aligned} & 0 \leq p \leq q \leq n + 1 \\ &= 0 \leq p \leq q \wedge q \leq n + 1 \\ &= 0 \leq p \leq q \wedge (q \leq n \vee q = n + 1) \\ &= (0 \leq p \leq q \wedge q \leq n) \vee (0 \leq p \leq q \wedge q = n + 1) \\ &= 0 \leq p \leq q \leq n \vee (0 \leq p \leq q \wedge q = n + 1) . \end{aligned}$$

Without information about n , nothing guarantees that the ranges $0 \leq p \leq q \leq n$ and $0 \leq p \leq q \wedge q = n + 1$ are not empty. It does not matter yet, *for now*.

Therefore we have:

$$\begin{aligned}
 & \langle \uparrow p \ q : 0 \leq p \leq q \leq n + 1 : \text{sum } p \ q \rangle \\
 = & \quad \{ \text{previous calculation} \} \\
 & \langle \uparrow p \ q : 0 \leq p \leq q \leq n \vee \\
 & \quad (0 \leq p \leq q \wedge q = n + 1) : \text{sum } p \ q \rangle \\
 = & \quad \{ \text{range split (8.16)} \} \\
 & \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle \uparrow \\
 & \quad \langle \uparrow p \ q : 0 \leq p \leq q \wedge q = n + 1 : \text{sum } p \ q \rangle \\
 = & \quad \{ \text{nesting (8.20)} \} \\
 & \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle \uparrow \\
 & \quad \langle \uparrow q : q = n + 1 : \langle \uparrow p : 0 \leq p \leq q : \text{sum } p \ q \rangle \rangle \\
 = & \quad \{ \text{one-point rule} \} \\
 & \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle \uparrow \\
 & \quad \langle \uparrow p : 0 \leq p \leq n + 1 : \text{sum } p \ (n + 1) \rangle .
 \end{aligned}$$

Things to note:

- Calculation for other patterns of ranges (e.g. $0 \leq p \leq q \leq n + 1$) are slightly different. Watch out!
- In practice, the “splitting off” step is but one quick step. We do not do the reasoning above in such detail.
- We show you the details above for expository purpose.
- In other problems we may see slightly different ranges, such as $0 \leq p < q < n + 1$. The result of splitting is different too. Take extra care!

STRENGTHENING THE INVARIANT

Knowing that we need to update r with $\langle \uparrow p : 0 \leq p \leq (n + 1) : \text{sum } p (n + 1) \rangle$, let us store it in some variable! Introduce a new variable s , and *strengthen* the invariant to $P_0 \wedge P_1 \wedge Q$, where

$$P_0 \equiv r = \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle ,$$

$$P_1 \equiv s = \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n \rangle ,$$

$$Q \equiv 0 \leq n \leq N .$$

MAXIMUM SUFFIX SUM

- That is, while r is the maximum *segment* sum so far, s is the maximum *suffix* sum so far.
- We discover the need of this concept through symbolic calculation.
- This is a pattern for many “segment problems”: *to solve a problem about segments, solve a suffix problem for all prefixes.*

Q: Why don't we let $s = \langle \uparrow p : 0 \leq p \leq n + 1 : \text{sum } p(n + 1) \rangle$?

A: For this example you will run into some problems. The details are left as an exercise. But in general it is not always a bad idea.

CONSTRUCTING THE LOOP BODY

Therefore, a possible strategy would be:

$$\{P_0 \wedge P_1 \wedge 0 \leq n \leq N \wedge n \neq N\}$$

$$s := ???$$

$$\{P_0 \wedge P_1[n \setminus n+1] \wedge 0 \leq n+1 \leq N\}$$

$$r := r \uparrow s$$

$$\{(P_0 \wedge P_1 \wedge 0 \leq n \leq N)[n \setminus n+1]\}$$

$$n := n + 1$$

$$\{P_0 \wedge P_1 \wedge 0 \leq n \leq N\}$$

UPDATING THE PREFIX SUM

Recall $P_1 \equiv s = \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n \rangle$.

$$\begin{aligned} & \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n \rangle [n \setminus n + 1] \\ &= \langle \uparrow p : 0 \leq p \leq n + 1 : \text{sum } p \ (n + 1) \rangle \\ &= \{ \text{splitting off } p = n + 1 \} \\ & \quad \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ (n + 1) \rangle \uparrow \\ & \quad \text{sum } (n + 1) \ (n + 1) \\ &= \{ [n + 1..n + 1] \text{ is an empty range} \} \\ & \quad \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ (n + 1) \rangle \uparrow 0 \\ &= \{ \text{splitting off } i = n \text{ in sum} \} \\ & \quad \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n + f[n] \rangle \uparrow 0 \\ &= \{ \text{distributivity} \} \\ & \quad (\langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n \rangle + f[n]) \uparrow 0 . \end{aligned}$$

Thus, $\{P_1\} s := ? \{P_1[n \setminus n + 1]\}$ is satisfied by $s := (s + f[n]) \uparrow 0$.

SPLITTING OFF — THINGS TO WATCH OUT

We look at the step “splitting off $i = n$ ” in detail. See the range calculation:

$$\begin{aligned} & p \leq i < n + 1 \\ &= p \leq i \wedge (i < n \vee i = n) \\ &= p \leq i < n \vee (p \leq i \wedge i = n) \\ &= \{ \text{we need } 0 \leq n! \} \\ & \quad p \leq i < n \vee i = n \end{aligned}$$

Compare this to the previous range calculation. This time we completely remove $p \leq i$.

It allows us to perform one-point rule, without nesting:

$$\begin{aligned} & \text{sum } p \ (n + 1) \\ &= \langle \Sigma i : p \leq i < n + 1 : f[i] \rangle \\ &= \{ \text{range calculation} \} \\ & \quad \langle \Sigma i : p \leq i < n \vee i = n : f[i] \rangle \\ &= \langle \Sigma i : p \leq i < n : f[i] \rangle + \langle \Sigma i : i = n : f[i] \rangle \\ &= \{ \text{one-point rule} \} \\ & \quad \langle \Sigma i : p \leq i < n : f[i] \rangle + f[n] \ . \end{aligned}$$

However, that means

- we need to reduce $p \leq i \wedge i = n$ to $i = n$.
- That is, $p \leq i$ does not put more constraints on $i = n$. In particular, $i = n$, when conjuncted with $p \leq i$, cannot reduce to *False*,
- or, $p \leq n$ cannot be an empty range.
- Since in the outer quantification we have $0 \leq p \leq n$, we need $0 \leq n$.

That is why we need $0 \leq n$ in the invariant!

Lesson: as long as the quantification is around, we do not care whether the range is empty. We do have to check that the range is not empty when the one-point rule leaves no remaining quantifications.

The requirement we need to ensure that the range is not empty are often added to the loop invariant.

A KEY PROPERTY

- The last step labelled “distributivity” uses a rule mentioned before: provided that $\neg occurs(i, F)$ and R non-empty:

$$F + \langle \uparrow i : R : S \rangle = \langle \uparrow i : R : F + S \rangle$$

$$F + \langle \downarrow i : R : S \rangle = \langle \downarrow i : R : F + S \rangle .$$

- The rules are valid because addition distributes into maximum/minimum:

$$x + (y \uparrow z) = (x + y) \uparrow (x + z) ,$$

$$x + (y \downarrow z) = (x + y) \downarrow (x + z) .$$

- That is the key property that allows us to have an efficient algorithm for the maximum segment sum problem!
- Through calculation, we not only have an algorithm, but also identified the key property that makes it work, which

DERIVED PROGRAM

```
con  $N : \text{Int } \{0 \leq N\}$   
con  $f : \text{array } [0..N) \text{ of } \text{Int}$   
var  $r, s, n : \text{Int}$   
  
 $r, s, n := 0, 0, 0$   
 $\{P_0 \wedge P_1 \wedge Q, bnd : N - n\}$   
do  $n \neq N \rightarrow$   
     $s := (s + f[n]) \uparrow 0$   
     $r := r \uparrow s$   
     $n := n + 1$   
od  
 $\{r = \langle \uparrow p \ q : 0 \leq p \leq q \leq N : \text{sum } p \ q \rangle\}$ 
```

$P_0 \equiv r = \langle \uparrow p \ q : 0 \leq p \leq q \leq n : \text{sum } p \ q \rangle$,

$P_1 \equiv s = \langle \uparrow p : 0 \leq p \leq n : \text{sum } p \ n \rangle$,

$Q \equiv 0 \leq n \leq N$.

“STRENGTHENING”?

- We say that the invariant $P_0 \wedge P_1 \wedge Q$ is “stronger” than $P \wedge Q$ because the former promises more.
- The resulting loop computes values for two variables rather than one.
- However, the program ends up being quicker because more results from the previous iteration of the loop can be utilised.
- It is a common phenomena: a generalised theorem is easier to prove.
- We will see another way to generalise the invariant in the rest of the course.

LESSONS LEARNT?

Let the symbols do the work!

- We discover how to strengthen the invariant by calculating and finding out what is missing.
- Expressions are your friend, and blind guessing can be minimised. We always get some clue from the expressions.
- Since we rely only on the symbols, the same calculation/algorithm can be generalised to other problems (e.g. as long as the same distributivity property holds).

If we remove the pre/postconditions and the invariant, can you tell us what the program does?

- Without the assertions, programs mean nothing. The assertions are what matter about the program.
- Structured programming is not about making (the operational parts of) code easier to read/understand.
- Such efforts are bound to end in vain: even a simple three-line loop can be hard to understand if the assertions, encoding the intentions of the programmer, are stripped away.

- Instead, structured programming is about organising the code around the structure of the proofs.
- Once the pre/postconditions are given, and the invariants and bounds are determined, one can derive the code accordingly.
- It is pointless arguing, for example, “using a *break* here makes the code easier to read.”
- One shall not need to “understand” the operational parts of the code, but to check whether it meets the specification.

NO. OF PAIRS IN AN ARRAY

Consider constructing the following program:

```
con  $N : \text{Int}$   $\{0 \leq N\}$ ;  $a : \text{array } [0..N)$  of  $\text{Int}$ 
```

```
var  $r : \text{Int}$ 
```

```
 $S$ 
```

```
 $\{r = \langle \#i\ j : 0 \leq i < j < N : a[i] \leq 0 \wedge a[j] \geq 0 \rangle\}$ 
```

PREVIOUSLY INTRODUCED TECHNIQUES

- Replace N by n . Use $P \wedge Q$ as the invariant, where

$$P \equiv r = \langle \#i, j : 0 \leq i < j < n : a[i] \leq 0 \wedge a[j] \geq 0 \rangle,$$

$$Q \equiv 0 \leq n \leq N.$$

- Use $\neg (n = N)$ as guard. This way we immediately have that $P \wedge Q \wedge n = N$ imply the desired postcondition.
- Initialisation: $n, r := 0, 0$.
- Use $N - n$ as the bound.
- To decrease the bound, let $n := n + 1$ be the last statement of the loop.

We get this program.

```
con  $N : \text{Int } \{0 \leq N\}; a : \text{array } [0..N) \text{ of } \text{Int}$   
var  $r, n : \text{Int}$   
 $r, n := 0, 0$   
 $\{P \wedge Q, \text{bnd} : N - n\}$   
do  $n \neq N \rightarrow \dots; n := n + 1$  od  
 $\{r = \langle \#i\ j : 0 \leq i < j < N : a[i] \leq 0 \wedge a[j] \geq 0 \rangle\}$ 
```

Now we need to construct the ... part.

How to construct the ... part?

$$\{P \wedge Q \wedge n \neq N\}$$
$$\dots$$
$$\{(P \wedge Q)[n \setminus n + 1]\}$$
$$n := n + 1$$
$$\{P \wedge Q\}$$

NO. OF PAIRS IN AN ARRAY

To reason about $P[n \setminus n + 1]$, we calculate (assuming $P \wedge Q \wedge n \neq N$):

$$\begin{aligned} & \langle \#i, j : 0 \leq i < j < n + 1 : a[i] \leq 0 \wedge a[j] \geq 0 \rangle \\ = & \{ \text{split off } j = n, \text{ see the next slide} \} \\ & \langle \#i, j : 0 \leq i < j < n : a[i] \leq 0 \wedge a[j] \geq 0 \rangle + \\ & \langle \#i : 0 \leq i < n : a[i] \leq 0 \wedge a[n] \geq 0 \rangle \\ = & \{ P \} \\ & r + \langle \#i : 0 \leq i < n : a[i] \leq 0 \wedge a[n] \geq 0 \rangle \\ = & \begin{cases} r, & \text{if } a[n] < 0; \\ r + \langle \#i : 0 \leq i < n : a[i] \leq 0 \rangle, & \text{if } a[n] \geq 0. \end{cases} \end{aligned}$$

Let us try storing $\langle \#i : 0 \leq i < n : a[i] \leq 0 \rangle$ in another variable?

SPLITTING OFF?

For expository purpose let us exam how the splitting was done:

$$\begin{aligned} & 0 \leq i < j < n + 1 \\ &= 0 \leq i < j \wedge j < n + 1 \\ &= 0 \leq i < j \wedge (j < n \vee j = n) \\ &= (0 \leq i < j \wedge j < n) \vee (0 \leq i < j \wedge j = n) \\ &= 0 \leq i < j < n \vee (0 \leq i < j \wedge j = n) . \end{aligned}$$

Without information on n , either of the ranges could be empty.

A FREQUENT PATTERN

We may see this pattern often. For some \star , we need to calculate:

$$\begin{aligned} & \langle \star i j : 0 \leq i < j < n + 1 : R \rangle \\ = & \{ \text{previous calculation} \} \\ & \langle \star i j : 0 \leq i < j < n \vee (0 \leq i < j \wedge j = n) : R \rangle \\ = & \langle \star i j : 0 \leq i < j < n : R \rangle \star \\ & \langle \star i j : 0 \leq i < j \wedge j = n : R \rangle \\ = & \{ \text{nesting (8.20)} \} \\ & \langle \star i j : 0 \leq i < j < n : R \rangle \star \\ & \langle \star j : j = n : \langle \star i : 0 \leq i < j : R \rangle \rangle \\ = & \{ \text{one-point rule} \} \\ & \langle \star i j : 0 \leq i < j < n : R \rangle \star \\ & \langle \star i : 0 \leq i < n : R[j \setminus n] \rangle . \end{aligned}$$

Calculation for other ranges (e.g. $0 \leq i \leq j \leq n + 1$) are slightly different. Watch out!

STRENGTHENING THE INVARIANT

New plan: define

$$P_0 \equiv r = \langle \#i, j : 0 \leq i < j < n : a[i] \leq 0 \wedge a[j] \geq 0 \rangle,$$

$$P_1 \equiv s = \langle \#i : 0 \leq i < n : a[i] \leq 0 \rangle,$$

$$Q \equiv 0 \leq n \leq N,$$

and try to derive

```
con  $N : \text{Int}$  { $N \geq 0$ };  $a : \text{array}[0..N)$  of  $\text{Int}$   
var  $n, r, s : \text{Int}$ 
```

```
 $n, r, s := 0, 0, 0$ 
```

```
{ $P_0 \wedge P_1 \wedge Q, \text{bnd} : N - n$ }
```

```
do  $n \neq N \rightarrow \dots n := n + 1$  od
```

```
{ $r = \langle \#i, j : 0 \leq i < j < N : a[i] \leq 0 \wedge a[j] \geq 0 \rangle$ }
```

UPDATE THE NEW VARIABLE

$$\begin{aligned} & \langle \#i : 0 \leq i < n : a[i] \leq 0 \rangle [n \setminus n + 1] \\ = & \langle \#i : 0 \leq i < n + 1 : a[i] \leq 0 \rangle \\ = & \{ \text{split off } i = n \text{ (assuming } 0 \leq n) \} \\ & \langle \#i : 0 \leq i < n : a[i] \leq 0 \rangle + \#(a[n] \leq 0) \\ = & \{ P_1 \} \\ & s + \#(a[n] \leq 0) \\ = & \begin{cases} s & \text{if } a[n] > 0, \\ s + 1 & \text{if } a[n] \leq 0. \end{cases} \end{aligned}$$

RESULTING PROGRAM

```
 $n, r, s := 0, 0, 0$   
 $\{P_0 \wedge P_1 \wedge Q, bnd : N - n\}$   
do  $n \neq N \rightarrow \{P_0 \wedge P_1 \wedge Q \wedge n \neq N\}$   
  if  $a[n] < 0 \rightarrow skip$   
     $| a[n] \geq 0 \rightarrow r := r + s$   
  fi  
   $\{P_0[n \setminus n + 1] \wedge P_1 \wedge Q \wedge n \neq N\}$   
  if  $a[n] > 0 \rightarrow skip$   
     $| a[n] \leq 0 \rightarrow s := s + 1$   
  fi  
   $\{(P_0 \wedge P_1 \wedge Q)[n \setminus n + 1]\}$   
   $n := n + 1$   
od  
 $\{r = \langle \#i, j : 0 \leq i < j < N : a[i] \leq 0 \wedge a[j] \geq 0 \rangle\}$ 
```

RESULTING PROGRAM

Since $P_0 \wedge P_1 \wedge Q \wedge n \neq N$ is a common precondition for the **if**'s (the second **if** does not use P_0), they can be combined:

```
 $n, r, s := 0, 0, 0$   
 $\{P_0 \wedge P_1 \wedge Q, bnd : N - n\}$   
do  $n \neq N \rightarrow \{P_0 \wedge P_1 \wedge Q \wedge n \neq N\}$   
  if  $a[n] < 0 \rightarrow s := s + 1$   
     $| a[n] = 0 \rightarrow r, s := r + s, s + 1$   
     $| a[n] > 0 \rightarrow r := r + s$   
  fi  
   $\{(P_0 \wedge P_1 \wedge Q)[n \setminus n + 1]\}$   
   $n := n + 1$   
od  
 $\{r = \langle \#i, j : 0 \leq i < j < N : a[i] \leq 0 \wedge a[j] \geq 0 \rangle\}$ 
```

However, from the point of view of program derivation, the first program is totally fine.

It closely matches the structure of proofs. If one tries to understand a program by how its proof proceeds (which is the way a program should be understood), rather than trying to read it operationally, one may argue that first program is easier to understand.

ISN'T IT GETTING A BIT TOO COMPLICATED?

- Quantifier and indexes manipulation tend to get very long and tedious.
 - Expect to see even longer expressions later!
- To certain extent, it is a restriction of the data structure we are using. With arrays we have to manipulate the indexes.
- Is it possible to use higher-level data structures? Lists? Trees?
 - Heap-allocated data structure with pointers is a horrifying beast!
 - Trying to be more abstract lead to further developments in programming languages, e.g. algebraic datatypes.